Example 5b: General Linear Models with Multiple Fixed Effects of Multiple Predictors Simultaneously in SAS and STATA (with Hoffman 2015 Chapter 2 example data)

The models for this example come from Hoffman (2015) chapter 2. We will be examining the extent to which cognition (as measured by an information test outcome) can be predicted from age (centered at 85 years), grip strength (centered at 9 pounds), sex (with men as the reference group) and subsequent dementia status (none = 1, future = 2, and current = 3, with none as the reference) in a sample of 550 older adults.

SAS Syntax for Importing and Preparing Data for Analysis:

```
* Defining global variable for file location to be replaced in code below;
* \\Client\ precedes path in Virtual Desktop outside H drive;
  %LET filesave= C:\Dropbox\21SP_PSQF6242\PSQF6242_Example5b;
* Location for SAS files for these models (uses macro variable filesave);
  LIBNAME filesave "&filesave.";

* Import chapter 2 example data into work library as Example5b;
DATA work.Example5b; SET filesave.SAS_Chapter2;
* Center quantitative predictors;
  age85 = age - 85;
  grip9 = grip - 9;
* Create dummy-coded binary predictors for dementia groups;
  demNF=.; demNC=.; * Create two new empty variables;
  IF demgroup=1 THEN DO; demNF=0; demNC=0; END; * Replace each for none group;
  IF demgroup=2 THEN DO; demNF=1; demNC=0; END; * Replace each for future group;
  IF demgroup=3 THEN DO; demNF=0; demNC=1; END; * Replace each for current group;
* Label new variables – note semi-colon is only at the end of ALL labels;
  LABEL
  age85=     "age85: Age in Years (0=85)"
  grip9=     "grip9: Grip Strength in Pounds (0=9)"
  sexMW=     "sexMW: Sex (0=M, 1=W)"
  demNF=     "demNF: Dementia Contrast for None=0 vs Future=1"
  demNC=     "demNC: Dementia Contrast for None=0 vs Current=1"
  cognition= "cognition: Cognition Outcome"
  demgroup=  "demgroup: Dementia Group 1N 2F 3C";
* Select cases complete on variables;
  IF NMISS(cognition,age,grip,sexmw,demgroup)>0 THEN DELETE;
RUN;
```

STATA Syntax for Importing and Preparing Data for Analysis:

```
// Defining global variable for file location to be replaced in code below
// \\Client\ precedes path in Virtual Desktop outside H drive
   global filesave "C:\Dropbox\21SP_PSQF6242\PSQF6242_Example5b"

// Import chapter 2 data into temporary file and center predictors
   use "$filesave\STATA_Chapter2.dta", clear // Has converted all variables to lower-case

// Center quantitative predictors
   gen age85 = age - 85
   gen grip9 = grip - 9
// Create dummy-coded binary predictors for dementia groups
   gen demnf=.  // Create two new empty variables
   gen demnc=.
// Replace for demgroup = none
   replace demnf=0 if demgroup==1
   replace demnc=0 if demgroup==1
// Replace for demgroup = future
   replace demnf=1 if demgroup==2
   replace demnc=0 if demgroup==2
// Replace for demgroup = current
   replace demnf=0 if demgroup==3
   replace demnc=1 if demgroup==3
```

```
// Label all variables
   label variable age85     "age85: Age in Years (0=85)"
   label variable grip9     "grip9: Grip Strength in Pounds (0=9)"
   label variable sexmw     "sexmw: Sex (0=Men, 1=Women)"
   label variable demnf     "demnf: Dementia Contrast for None=0 vs Future=1"
   label variable demnc     "demnc: Dementia Contrast for None=0 vs Current=1"
   label variable cognition "cognition: Cognition Outcome"
   label variable demgroup  "demgroup: Dementia Group 1N 2F 3C"
// Select cases complete on variables
   egen nmiss=rowmiss(cognition age grip sexmw demgroup)
   drop if nmiss>0
```

Syntax and SAS Output for Creating Descriptive Statistics:

```
TITLE "SAS Descriptive Statistics";
PROC MEANS NOLABELS NONOBS NDEC=3 MEAN STDDEV VAR MIN MAX DATA=work.Example5b;
     VAR cognition age grip sexMW;
RUN;

display "STATA Descriptive Statistics"
format cognition age grip sexmw demnf demnc %4.3f
summarize cognition age grip sexmw demnf demnc, format
summarize cognition age grip sexmw demnf demnc, detail
```

SAS Descriptive Statistics

Variable	Mean	Std Dev	Variance	Minimum	Maximum
cognition	24.822	10.989	120.759	0.000	44.000
age	84.927	3.430	11.765	80.016	96.967
grip	9.113	2.983	8.898	0.000	19.000
sexMW	0.587	0.493	0.243	0.000	1.000

```
PROC FREQ DATA=work.Example5b;
     TABLE sexMW*demgroup / NOROW NOCOL;
RUN; TITLE;

display "STATA Descriptive Statistics"
tabulate sexmw demgroup
```

sexMW(sexMW: Sex (0=M, 1=W))
demgroup: demgroup(Dementia Group 1N 2F 3C)

Frequency / Percent	1	2	3	Total
0	168 30.55	40 7.27	19 3.45	227 41.27
1	231 42.00	69 12.55	23 4.18	323 58.73
Total	399 72.55	109 19.82	42 7.64	550 100.00

```
TITLE "SAS Histograms of Quantitative Variables";
PROC UNIVARIATE NOPRINT DATA=work.Example5b;
     VAR cognition age grip ;
     HISTOGRAM age / MIDPOINTS=80 TO 97 BY 1;
     HISTOGRAM grip / MIDPOINTS=0 TO 19 BY 1;
     HISTOGRAM cognition / MIDPOINTS=0 TO 45 BY 5;
RUN; QUIT; TITLE;

display "STATA Histograms of Quantitative Variables"
histogram age,       percent discrete width(1) start(80)
histogram grip,      percent discrete width(1) start(0)
histogram cognition, percent discrete width(5) start(0)
```







The cognition outcome appears relatively normally distributed.

Here is what the residuals look like after the final model:




```
TITLE "SAS Bivariate Correlations (OUT saves corrs to dataset)";
PROC CORR NOSIMPLE DATA=work.Example5b OUT=work.Corrs;
     VAR cognition age grip sexMW;
RUN; TITLE;

display "STATA Bivariate Correlations"
pwcorr cognition age grip sexmw, sig
```

Note that the binary contrasts for dementia group are not included because their meaning is different for each in separate correlations than when they both are together in the same model.

Pearson Correlation Coefficients, N = 550
Prob > |r| under H0: Rho=0 **(significant correlations in bold)**

	cognition	age	grip	sexMW
cognition Cognition Outcome	1.00000	**-0.17045** <.0001	**0.24183** <.0001	**-0.23628** <.0001
age age: Age in Years	**-0.17045** <.0001	1.00000	**-0.18414** <.0001	0.04560 0.2858
grip grip: Grip Strength in Pounds	**0.24183** <.0001	**-0.18414** <.0001	1.00000	**-0.40324** <.0001
sexMW sexMW: Sex (0=M, 1=W)	**-0.23628** <.0001	0.04560 0.2858	**-0.40324** <.0001	1.00000

What can we conclude about the linear relationships between each pair of variables?

Cognition and age:

Cognition and grip:

Cognition and sex:

Age and grip:

Age and sex:

Grip and sex:

Syntax and SAS Output for General Linear Model Predicting Cognition as Outcome:

$$Cognition_i = \beta_0 + e_i$$

```
TITLE "SAS Empty Model Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
     MODEL cognition = / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
RUN; QUIT; TITLE;

display "STATA Empty Model Predicting Cognition"
regress cognition, level(95)
```

In an empty model, the DF numerator includes the mean for the model SS ("uncorrected total"), but not after adding predictors ("corrected total").

SAS Empty Model for Cognition Outcome

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	338867.4618	338867.4618	2806.15	<.0001
Error	549	66296.5382	**120.7587**		
Uncorrected Total	550	405164.0000			

R-Square	Coeff Var	Root MSE	cognition Mean
0.000000	44.27165	10.98903	24.82182

Mean Square Error (Mean Square **Residual** in STATA) gives the residual variance = 120 here. In the empty model this is ALL the outcome variance.

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	24.82181818	0.46857370	52.97	<.0001	23.90140147	25.74223490	**Beta0**

Syntax and SAS Output with Linear Age (0=85 years) as Predictor of Cognition:

$$Cognition_i = \beta_0 + \beta_1(Age_i - 85) + e_i$$

```
TITLE "SAS Linear Age (0=85) Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
     MODEL cognition = age85 / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
     ODS OUTPUT ParameterEstimates=AgeLinEstimates; * Save for computing effect sizes;
RUN; QUIT; TITLE;

display "STATA Linear Age (0=85) Predicting Cognition"
regress cognition c.age85, level(95)
```

In STATA c. indicates a quantitative predictor (c. is needed to do math on them, like quadratic terms below)

SAS Linear Age (0=85) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	**1926.18133**	1926.18133	**16.40**	<.0001
Error	548	64370.35685	**117.46415**		
Corrected Total	549	**66296.53818**			

R-Square	Coeff Var	Root MSE	cognition Mean
0.029054	43.66355	10.83809	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
age85	1	**1926.181332**	1926.181332	16.40	<.0001

Because there is only one slope, the F for the model is the same as t^2 for the age slope. The model SS is the same as the age slope SS, and the model R^2 is the same as the squared semi-partial correlation sr^2 ("semipartial eta-square") for the age slope.

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
age85	**0.0291**	0.0272	0.0078	0.0619

$$sr^2 = SS_{slope}/SS_{total}$$
$$sr^2 = \frac{1926.18}{66296.84} = .0291$$

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	24.78183688	0.46224309	53.61	<.0001	23.87385169	25.68982207	**Beta0**
age85	-0.54609053	0.13485552	-4.05	<.0001	-0.81098756	-0.28119351	**Beta1**

Interpret β_0 = Intercept:

Interpret β_1 = slope of age85:

```
* Compute Cohen d effect size and r effect sizes from saved solution;
DATA work.AgeLinEstimates; SET work.AgeLinEstimates; %LET df=548;
     CohenD=(2*tvalue)/SQRT(&df.);
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.);
RUN;

* Print effect size;
TITLE "Effect Size for Linear Age";
PROC PRINT NOOBS DATA=work.AgeLinEstimates;
     WHERE INDEX(Parameter,"Intercept")=0; * Do not print intercept;
     VAR Parameter Estimate StdErr probt CohenD EffectR;
RUN; TITLE;
```

Effect Sizes for Linear Age

Parameter	Estimate	StdErr	Probt	CohenD	EffectR	
age85	-0.54609053	0.13485552	<.0001	-0.34597	-0.17045	→ same as Pearson r

Syntax and SAS Output adding Quadratic Age (0=85 years) as Predictor of Cognition:

$$Cognition_i = \beta_0 + \beta_1(Age_i - 18) + \beta_2(Age_i - 18)^2 + e_i$$

```
TITLE "SAS Quadratic Age (0=85) Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
     MODEL cognition = age85 age85*age85 / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
     ODS OUTPUT ParameterEstimates=AgeQuadEstimates; * Save for computing effect sizes;
RUN; QUIT; TITLE;

display "STATA Quadratic Age (0=85) Predicting Cognition"
regress cognition c.age85 c.age85#c.age85, level(95)
pcorr cognition c.age85 c.age85#c.age85 // Get semi-partial eta-squared for each slope
```

SAS Quadratic Age (0=85) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	2	**1961.68396**	980.84198	**8.34**	0.0003
Error	547	64334.85422	**117.61399**		
Corrected Total	549	**66296.53818**			

R-Square	Coeff Var	Root MSE	cognition Mean
0.029590	43.69139	10.84500	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
age85	1	**1386.263193**	1386.263193	11.79	0.0006
age85*age85	1	**35.502630**	35.502630	0.30	0.5829

Because of covariance between the predictors (age and age^2), the model SS is greater than the sum of the slope SS values, and the model R^2 is greater than the sum of the slope sr^2 values.

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
age85	**0.0209**	0.0191	0.0038	0.0502
age85*age85	**0.0005**	-0.0012	0.0000	0.0111

$sr^2 = SS_{slope}/SS_{total}$

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	24.57135998	0.60058398	40.91	<.0001	23.39162668	25.75109327	**Beta0**
age85	-0.60946483	0.17752327	-3.43	0.0006	-0.95817562	-0.26075403	**Beta1**
age85*age85	0.01751944	0.03188742	0.55	0.5829	-0.04511735	0.08015623	**Beta2**

Interpret β_0 = Intercept:

Interpret β_1 = slope of age85:

Interpret β_2 = slope of age85^2:

Do we know if the R^2 changed significantly relative to the linear age model?

```
* Compute Cohen d effect size and r effect sizes from saved solution;
DATA work.AgeQuadEstimates; SET work.AgeQuadEstimates; %LET df=547;
     CohenD=(2*tvalue)/SQRT(&df.);
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.);
RUN;

* Print effect sizes;
TITLE "Effect Sizes for Linear and Quadratic Age";
PROC PRINT NOOBS DATA=work.AgeQuadEstimates;
     WHERE INDEX(Parameter,"Intercept")=0; * Do not print intercept;
     VAR Parameter Estimate StdErr probt CohenD EffectR;
RUN; TITLE;
```

```
Effect Sizes for Linear and Quadratic Age
Parameter        Estimate        StdErr  Probt  CohenD    EffectR
age85         -0.60946483    0.17752327 0.0006 -0.29358 -0.14523
age85*age85    0.01751944    0.03188742 0.5829  0.04698  0.02348 → unique controlling for linear age
```

Syntax and SAS Output with Linear Grip (0=9 pounds) as Predictor of Cognition:

$$Cognition_i = \beta_0 + \beta_1(Grip_i - 9) + e_i$$

```
TITLE "SAS Linear Grip (0=9) Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
     MODEL cognition = grip9 / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
     ODS OUTPUT ParameterEstimates=GripLinEstimates; * Save for computing effect sizes;
RUN; QUIT; TITLE;

display "STATA Linear Grip (0=9) Predicting Cognition"
regress cognition c.grip9, level(95)
```

SAS Linear Grip (0=9) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	**3877.23095**	3877.23095	**34.04**	<.0001
Error	548	62419.30724	**113.90385**		
Corrected Total	549	**66296.53818**			

R-Square	Coeff Var	Root MSE	cognition Mean
0.058483	42.99675	10.67257	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
grip9	1	**3877.230945**	3877.230945	34.04	<.0001

Because there is only one slope, the F for the model is the same as t^2 for the grip slope. The model SS is the same as the grip slope SS, and the model R^2 is the same as the squared semi-partial correlation sr^2 ("semipartial eta-square") for the grip slope.

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
grip9	**0.0585**	0.0567	0.0261	0.1001

$$sr^2 = SS_{slope}/SS_{total}$$
$$sr^2 = \frac{3877.23}{66296.84} = .0585$$

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits	
Intercept	24.72138966	0.45540553	54.28	<.0001	23.82683550	25.61594383
grip9	0.89089817	0.15269908	5.83	<.0001	0.59095100	1.19084533

Interpret β_0 = Intercept:

Interpret β_1 = slope of grip9:

```
* Compute Cohen d effect size and r effect sizes from saved solution;
DATA work.GripLinEstimates; SET work.GripLinEstimates; %LET df=548;
     CohenD=(2*tvalue)/SQRT(&df.);
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.); RUN;

* Print effect size;
TITLE "Effect Size for Linear Grip";
PROC PRINT NOOBS DATA=work.GripLinEstimates;
     WHERE INDEX(Parameter,"Intercept")=0; * Do not print intercept;
     VAR Parameter Estimate StdErr probt CohenD EffectR; RUN; TITLE;
```

```
Effect Size for Linear Grip
Parameter       Estimate         StdErr     Probt     CohenD     EffectR
  grip9       0.89089817     0.15269908     <.0001    0.49846     0.24183 → same as Pearson r
```

Syntax and SAS Output adding Quadratic Grip (0=9 pounds) as Predictor of Cognition:

$$Cognition_i = \beta_0 + \beta_1(Grip_i - 9) + \beta_2(Grip_i - 9)^2 + e_i$$

```
TITLE "SAS Quadratic Grip (0=9) Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
     MODEL cognition = grip9 grip9*grip9 / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
     ODS OUTPUT ParameterEstimates=GripQuadEstimates; * Save for computing effect sizes;
RUN; QUIT; TITLE;

display "STATA Quadratic Grip (0=9) Predicting Cognition"
regress cognition c.grip9 c.grip9#c.grip9, level(95)
pcorr cognition c.grip9 c.grip9#c.grip9 // Get semi-partial eta-squared for each slope
```

SAS Quadratic Grip (0=9) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	2	**3897.77368**	1948.88684	**17.08**	<.0001
Error	547	62398.76450	**114.07452**		
Corrected Total	549	**66296.53818**			

R-Square	Coeff Var	Root MSE	cognition Mean
0.058793	43.02895	10.68057	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
grip9	1	3838.869863	3838.869863	33.65	<.0001
grip9*grip9	1	20.542739	20.542739	0.18	0.6715

Because of covariance between the predictors (age and age²), the model SS is greater than the sum of the slope SS values, and the model R^2 is greater than the sum of the slope sr^2 values.

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
grip9	**0.0579**	0.0561	0.0257	0.0994
grip9*grip9	**0.0003**	-0.0014	0.0000	0.0097

$sr^2 = SS_{slope}/SS_{total}$

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	24.57890721	0.56607286	43.42	<.0001	23.46696445	25.69084997	**Beta0**
grip9	0.88761523	0.15300914	5.80	<.0001	0.58705779	1.18817266	**Beta1**
grip9*grip9	0.01606069	0.03784681	0.42	0.6715	-0.05828220	0.09040358	**Beta2**

Interpret β_0 = Intercept:

Interpret β_1 = slope of grip9:

Interpret β_2 = slope of grip9²:

Do we know if the R^2 changed significantly relative to the linear grip model?

```
* Compute Cohen d effect size and r effect sizes from saved solution;
DATA work.GripQuadEstimates; SET work.GripQuadEstimates; %LET df=547;
     CohenD=(2*tvalue)/SQRT(&df.);
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.);
RUN;

* Print effect sizes;
TITLE "Effect Sizes for Linear and Quadratic Grip";
PROC PRINT NOOBS DATA=work.GripQuadEstimates;
     WHERE INDEX(Parameter,"Intercept")=0; * Do not print intercept;
     VAR Parameter Estimate StdErr probt CohenD EffectR;
RUN; TITLE;
```

```
Effect Sizes for Linear and Quadratic Grip
Parameter         Estimate        StdErr     Probt     CohenD     EffectR
grip9           0.88761523    0.15300914    <.0001    0.49607    0.24074
grip9*grip9     0.01606069    0.03784681    0.6715    0.03629    0.01814 →unique controlling for linear grip
```

Syntax and SAS Output with Two-Category Gender as Predictor of Cognition:

$$Cognition_i = \beta_0 + \beta_1(SexMW_i) + e_i$$

Men Mean: $\hat{y}_M = \beta_0 + \beta_1(0) = \beta_0$ ← fixed effect #1

Women Mean: $\hat{y}_W = \beta_0 + \beta_1(1) = \beta_0 + \beta_1$ ← linear combination

Diff of Men vs Women: $(\beta_0 + \beta_1) - (\beta_0) = \beta_1$ ← fixed effect #2

```
TITLE "SAS Sex (M=0, W=1) Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
     MODEL cognition = sexMW / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
     ESTIMATE "Cognition Mean for Men"   intercept 1 sexMW 0; * B0;
     ESTIMATE "Cognition Mean for Women" intercept 1 sexMW 1; * B0+B1;
     ESTIMATE "Mean Diff: Men vs. Women"             sexMW 1; * B1;
     ODS OUTPUT Estimates=SexEstimates; * Save for computing effect sizes;
RUN; QUIT; TITLE;
```

```
display "STATA Sex (M=0, W=1) Predicting Cognition"
regress cognition c.sexmw, level(95)
lincom _cons*1 + c.sexmw*0  // Cognition Mean for Men: B0
lincom _cons*1 + c.sexmw*1  // Cognition Mean for Women: B0+B1
lincom           c.sexmw*1  // Mean Diff: Men vs. Women: B1
```

SAS Sex (M=0, W=1) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	**3701.36378**	3701.36378	**32.40**	<.0001
Error	548	62595.17440	**114.22477**		
Corrected Total	549	**66296.53818**			

R-Square	Coeff Var	Root MSE	cognition Mean
0.055830	43.05728	10.68760	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
sexMW	1	**3701.363784**	3701.363784	32.40	<.0001

Because there is only one slope, the F for the model is the same as t^2 for the sex slope. The model SS is the same as the sex slope SS, and the model R^2 is the same as the squared semi-partial correlation sr^2 ("semipartial eta-square") for the sex slope.

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
sexMW	**0.0558**	0.0540	0.0243	0.0968

$$sr^2 = SS_{slope}/SS_{total}$$
$$sr^2 = \frac{3701.36}{66296.84} = .0558$$

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	27.91629956	0.70936086	39.35	<.0001	26.52290036	29.30969876	**Beta0**
sexMW	-5.26924074	0.92565106	-5.69	<.0001	-7.08749930	-3.45098217	**Beta1**

Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits	
Cognition Mean for Men	27.9162996	0.70936086	39.35	<.0001	26.5229004	29.3096988
Cognition Mean for Women	22.6470588	0.59467391	38.08	<.0001	21.4789395	23.8151782
Mean Diff: Men vs. Women	-5.2692407	0.92565106	-5.69	<.0001	-7.0874993	-3.4509822

```
* Compute Cohen d effect size and r effect sizes from saved solution;
DATA work.SexEstimates; SET work.SexEstimates; %LET df=547;
     CohenD=(2*tvalue)/SQRT(&df.);
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.);
RUN;

TITLE "Effect Sizes for 2-Category Gender Group";
PROC PRINT NOOBS DATA=work.SexEstimates;
     WHERE INDEX(Parameter,"Cognition")=0; * Do not print intercepts;
     VAR Parameter Estimate StdErr probt CohenD EffectR;
RUN; TITLE;
```

```
Effect Sizes for 2-Category Gender Group
       Parameter           Estimate       StdErr     Probt    CohenD     EffectR
Mean Diff: Men vs. Women -5.2692407   0.92565106    <.0001   -0.48634   -0.23628 → same as Pearson r
```

Syntax and SAS Output with Three-Category Dementia as Predictor of Cognition:

$$Cognition = \beta_0 + \beta_1(DemNF_i) + \beta_2(DemNC_i) + e_i$$

None Mean: $\hat{y}_N = \beta_0 + \beta_1(0) + \beta_2(0) = \beta_0$ ← fixed effect #1

Future Mean: $\hat{y}_F = \beta_0 + \beta_1(1) + \beta_2(0) = \beta_0 + \beta_1$ ← linear combination

Current Mean: $\hat{y}_C = \beta_0 + \beta_1(0) + \beta_2(1) = \beta_0 + \beta_2$ ← linear combination

Diff of None vs. Future: $(\beta_0 + \beta_1) - (\beta_0) = \beta_1$ ← fixed effect #2

Diff of None vs. Current: $(\beta_0 + \beta_2) - (\beta_0) = \beta_2$ ← fixed effect #3

Diff of Future vs. Current: $(\beta_0 + \beta_2) - (\beta_0 + \beta_1) = \beta_2 - \beta_1$ ← linear combination

```
TITLE "SAS Dementia Group (Ref=None) Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
     MODEL cognition = demNF demNC / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
     CONTRAST "Fut vs Cur" demNF -1 demNC  1; * Demo to get SS for linear combination;
     ESTIMATE "Cognition Mean for None"     intercept 1 demNF  0 demNC  0; * B0;
     ESTIMATE "Cognition Mean for Future"   intercept 1 demNF  1 demNC  0; * B0+B1;
     ESTIMATE "Cognition Mean for Current"  intercept 1 demNF  0 demNC  1; * B0+B2;
     ESTIMATE "Mean Diff: None vs. Future"              demNF  1 demNC  0; * B1;
     ESTIMATE "Mean Diff: None vs. Current"             demNF  0 demNC  1; * B2;
     ESTIMATE "Mean Diff: Future vs. Current"           demNF -1 demNC  1; * B2-B1;
     ODS OUTPUT Estimates=DemEstimates; * Save for computing effect sizes;
RUN; QUIT; TITLE;

display "STATA Dementia Group (Ref=None) Predicting Cognition"
regress cognition c.demnf c.demnc, level(95)
lincom _cons*1 + c.demnf*0 + c.demnc*0 // Cognition Mean for None    = B0
lincom _cons*1 + c.demnf*1 + c.demnc*0 // Cognition Mean for Future  = B0+B1
lincom _cons*1 + c.demnf*0 + c.demnc*1 // Cognition Mean for Current = B0+B2
lincom c.demnf*1  + c.demnc*0 // Mean Diff: None vs. Future    = B1
lincom c.demnf*0  + c.demnc*1 // Mean Diff: None vs. Current   = B2
lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B2-B1
```

SAS Dementia Group (Ref=None) Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	2	**11685.51093**	5842.75547	**58.52**	<.0001
Error	547	54611.02725	**99.83734**		
Corrected Total	549	**66296.53818**			

R-Square	Coeff Var	Root MSE	cognition Mean
0.176261	40.25436	9.991864	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
demNF	1	**2757.25209**	2757.25209	27.62	<.0001
demNC	1	**10206.11553**	10206.11553	102.23	<.0001
Fut vs Cur	1	**3479.801348**	3479.801348	34.85	<.0001

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
demNF	**0.0416**	0.0400	0.0150	0.0787
demNC	**0.1539**	0.1522	0.1027	0.2083
Fut vs Cur	**0.0525**	0.0509	0.0220	0.0927

Because the dementia group predictors are related (each shares a reference group with another), the total of the SS values for these three slopes is greater than the SS for the model (and the total of the sr^2 values for these slopes is greater than the R^2 for the model). For this reason, the predictor-specific sr^2 values should not be reported; only the overall model R^2 for *any* two slopes (that can then create the third slope) should be reported.

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	27.19799499	0.50021886	54.37	<.0001	26.21540992	28.18058005	**Beta0**
demNF	-5.67505921	1.07988789	-5.26	<.0001	-7.79629411	-3.55382430	**Beta1**
demNC	-16.38847118	1.62089436	-10.11	<.0001	-19.57241068	-13.20453167	**Beta2**

Interpret β_0 = Intercept:

Interpret β_1 = slope of None vs. Future:

Interpret β_2 = slope of None vs. Current:

Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects

Parameter	Estimate	Standard Error	t Value	Pr > |t|
Cognition Mean for None	27.1979950	0.50021886	54.37	<.0001
Cognition Mean for Future	21.5229358	0.95704699	22.49	<.0001
Cognition Mean for Current	10.8095238	1.54177807	7.01	<.0001
Mean Diff: None vs. Future	-5.6750592	1.07988789	-5.26	<.0001
Mean Diff: None vs. Current	-16.3884712	1.62089436	-10.11	<.0001
Mean Diff: Future vs. Current	-10.7134120	1.81466762	-5.90	<.0001

```
* Compute Cohen d effect size and r effect sizes;
DATA work.DemEstimates; SET work.DemEstimates; %LET df=547;
     CohenD=(2*tvalue)/SQRT(&df.); * Number = DF denominator;
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.);
RUN;

* Print effect sizes;
TITLE "Effect Sizes for 3-Category Dementia Status";
PROC PRINT NOOBS DATA=work.DemEstimates;
     VAR Parameter Estimate StdErr probt CohenD EffectR;
RUN; TITLE;
```

Effect Sizes for 3-Category Dementia Status (note: r values are NOT same as Pearson r values)

Parameter	Estimate	StdErr	Probt	CohenD	EffectR
Mean Diff: None vs. Future	-5.6750592	1.07988789	<.0001	-0.44939	-0.21923
Mean Diff: None vs. Current	-16.3884712	1.62089436	<.0001	-0.86461	-0.39681
Mean Diff: Future vs. Current	-10.7134120	1.81466762	<.0001	-0.50486	-0.24475

<u>Syntax and SAS Output with All Predictors of Cognition</u>

***(with requested F-tests to demonstrate how to test the unique joint contributions of 2+ slopes simultaneously, as in "hierarchical" or "stepwise" regression)*:**

$$Cognition_i = \beta_0 + \beta_1(Age_i - 85) + \beta_2(Grip_i - 9) + \beta_3(SexMW_i) + \beta_4(DemNF_i) + \beta_5(DemNC_i) + e_i$$

```
TITLE "SAS Combined Final Model Predicting Cognition";
PROC GLM DATA=work.Example5b NAMELEN=100;
MODEL cognition = age85 grip9 sexMW demNF demNC / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
CONTRAST "F-Test of DFnum=5 for Model"  age85 1, grip9 1, sexMW 1, demNF 1, demNC 1;
CONTRAST "F-Test DFnum=2 for Demgroup"      demNF  1, demNC  1; * For overall group;
CONTRAST "F-Test DFnum=2 for Age and Sex"   age85  1, sexMW  1; * For demographic vars;
ESTIMATE "Mean Diff: None vs. Future"       demNF  1  demNC  0; * B4;
ESTIMATE "Mean Diff: None vs. Current"      demNF  0  demNC  1; * B5;
ESTIMATE "Mean Diff: Future vs. Current"    demNF -1  demNC  1; * B5-B4;
ODS OUTPUT ParameterEstimates=ModelEstimates Estimates=ReqEstimates;
RUN; QUIT; TITLE; * Save for computing effect sizes;

display "STATA Combined Model Predicting Cognition"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc, level(95)
test (c.age85=0)(c.grip9=0)(c.sexmw=0)(c.demnf=0)(c.demnc=0) // F-Test DFnum=5 for Model
test (c.demnf=0)(c.demnc=0)   // F-Test DFnum=2 for Demgroup (for overall group)
test (c.age85=0)(c.sexmw=0)   // F-Test DFnum=2 for Age and Sex (for demographic vars)
lincom c.demnf*1  + c.demnc*0 // Mean Diff: None vs. Future    = B4
lincom c.demnf*0  + c.demnc*1 // Mean Diff: None vs. Current   = B5
lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B5-B4
// Get semipartial eta-squared for age, grip, and sex (dem slopes not valid)
pcorr cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc
```

SAS Combined Model Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	**18385.97930**	3677.19586	**41.75**	<.0001
Error	544	47910.55888	**88.07088**		
Corrected Total	549	**66296.53818**			

R-Square	Coeff Var	Root MSE	cognition Mean
0.277329	37.80790	9.384609	24.82182

Source	DF	Type III SS	Mean Square	F Value	Pr > F
age85	1	1025.58587	1025.58587	11.65	0.0007
grip9	1	1433.33570	1433.33570	16.27	<.0001
sexMW	1	1482.49792	1482.49792	16.83	<.0001
demNF	1	2776.56843	2776.56843	31.53	<.0001
demNC	1	10315.20016	10315.20016	117.12	<.0001

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits		
age85	0.0155	0.0141	0.0017	0.0418	
grip9	0.0216	0.0203	0.0041	0.0512	
sexMW	0.0224	0.0210	0.0044	0.0523	
demNF	0.0419	0.0405	0.0152	0.0791	→ Do not report (not valid)
demNC	0.1556	0.1541	0.1041	0.2100	→ Do not report (not valid)

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F	
F-test of DFnum=5 for Model	5	**18385.97930**	3677.19586	41.75	<.0001	→ for model
F-Test DFnum=2 for Demgroup	2	**11811.30155**	5905.65077	67.06	<.0001	→ for demgroup
F-Test DFnum=2 for Age and Sex	2	**2429.21021**	1214.60511	13.79	<.0001	→ for demog vars

Total Variation Accounted For

Contrast	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits		
F-Test DFnum=5 for Model (demo)	**0.2773**	0.2703	0.2116	0.3299	← 18385.98/66296.54
F-Test DFnum=2 for Demgroup	**0.1782**	0.1753	0.1227	0.2325	← 11811.30/66296.54
F-Test DFnum=2 for Age and Sex	**0.0366**	0.0339	0.0106	0.0702	← 2429.21/66296.54

```
// Stata: calculate semipartial eta-squared for demgroup using SS from reduced model
display "STATA Reduced Model to Get SS for demnf and demnc (not included)"
regress cognition c.age85 c.grip9 c.sexmw, level(95)
// sr2 = (SSfull-SSreduced)/SStotal
display (18385.97930-6574.67775)/66296.5382 // sr2 for demgroup = .17815865

// Stata: calculate semipartial eta-squared for age and sex using SS from reduced model
display "STATA Reduced Model to Get SS for age85 and sexmw (not included)"
regress cognition c.grip9 c.demnf c.demnc, level(95)
// sr2 = (SSfull-SSreduced)/SStotal
display (18385.97930-15956.7691)/66296.5382 // sr2 for age and sex = .03664158
```

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > |t|	95% Confidence Limits		
Intercept	29.26432541	0.69850792	41.90	<.0001	27.89222232	30.63642850	**Beta0**
age85	-0.40573396	0.11889717	-3.41	0.0007	-0.63928775	-0.17218017	**Beta1**
grip9	0.60422556	0.14977568	4.03	<.0001	0.31001605	0.89843507	**Beta2**
sexMW	-3.65737421	0.89143262	-4.10	<.0001	-5.40844590	-1.90630252	**Beta3**
demNF	-5.72197100	1.01907848	-5.61	<.0001	-7.72378184	-3.72016016	**Beta4**
demNC	-16.47981327	1.52275357	-10.82	<.0001	-19.47101037	-13.48861616	**Beta5**

Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects

Parameter	Estimate	Standard Error	t Value	Pr > |t|
Mean Diff: None vs. Future	-5.7219710	1.01907848	-5.61	<.0001
Mean Diff: None vs. Current	-16.4798133	1.52275357	-10.82	<.0001
Mean Diff: Future vs. Current	-10.7578423	1.70795708	-6.30	<.0001

```
* Compute Cohen d effect size and r effect sizes;
DATA work.FinalEstimates; LENGTH Parameter $50;
     SET work.ModelEstimates ReqEstimates; %LET df=544;
     CohenD=(2*tvalue)/SQRT(&df.);
     EffectR=tvalue/SQRT((tvalue*tvalue)+&df.); RUN;

* Print effect sizes;
TITLE "Effect Sizes for Final Model";
PROC PRINT NOOBS DATA=work.FinalEstimates;
     * Do not print intercept or redundant info;
     WHERE INDEX(Parameter,"Intercept")=0 AND INDEX(Parameter,"dem")=0;
     VAR Parameter Estimate StdErr probt CohenD EffectR; RUN; TITLE;
```

Effect Sizes for Final Model

Parameter	Estimate	StdErr	Probt	CohenD	EffectR
age85	-0.40573396	0.11889717	0.0007	-0.29262	-0.14477
grip9	0.60422556	0.14977568	<.0001	0.34593	0.17043
sexMW	-3.65737421	0.89143262	<.0001	-0.35181	-0.17325
Mean Diff: None vs. Future	-5.72197100	1.01907848	<.0001	-0.48147	-0.23405
Mean Diff: None vs. Current	-16.47981327	1.52275357	<.0001	-0.92801	-0.42090
Mean Diff: Future vs. Current	-10.75784226	1.70795708	<.0001	-0.54011	-0.26071

Syntax and SAS Output to Get Standardized Effects with All Predictors of Cognition:

```
TITLE "SAS Combined Model Predicting Cognition: Get Std Slopes";
PROC REG DATA=work.Example5b;
     MODEL cognition = age grip sexMW demNF demNC / ALPHA=.05 STB;
     * Would have to recode dem contrasts to get demFC std slope;
RUN; QUIT; TITLE;

display "STATA Combined Model Predicting Cognition -- Get Std Slopes"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc, beta
// would have to recode dem contrasts to get demfc std slope
```

SAS Combined Model Predicting Cognition: Get Std Slopes

Root MSE	9.38461	R-Square	0.2773
Dependent Mean	24.82182	Adj R-Sq	0.2707
Coeff Var	37.80790		

Parameter Estimates

Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > |t|	Standardized Estimate
Intercept	1	58.31368	10.49238	5.56	<.0001	0
age	1	-0.40573	0.11890	-3.41	0.0007	-0.12664
grip	1	0.60423	0.14978	4.03	<.0001	0.16402
sexMW	1	-3.65737	0.89143	-4.10	<.0001	-0.16400
demNF	1	-5.72197	1.01908	-5.61	<.0001	-0.20775
demNC	1	-16.47981	1.52275	-10.82	<.0001	-0.39864

The model is the same, but SAS REG (or STATA REGRESS, BETA) adds standardized slopes for model parameters, but not for linear combinations. To get standardized versions of linear combinations of fixed effects (such as the model-implied group difference between future and current dementia), we need to standardize all model variables (i.e., by z-scoring them), and then use those versions in PROC GLM or STATA REGRESS.

```
* SAS: Make a copy of variables to z-score for demonstration of standardized slopes;
DATA work.Example5b; SET work.Example5b;
     cognitionz=cognition; agez=age; gripz=grip;
     sexMWz=sexMW; demNFz=demNF; demNCz=demNC;
RUN;

* SAS: Z-score variables for demonstration of standardized slopes;
PROC STANDARD DATA= work.Example5b MEAN=0 STD=1 OUT=work.Example5b;
     VAR cognitionz agez gripz sexMWz demNFz demNCz;
RUN;

// STATA: Z-score variables for demonstration of standardized slopes
   egen cognitionz=std(cognition)
   egen agez=std(age)
   egen gripz=std(grip)
   egen sexmwz=std(sexmw)
   egen demnfz=std(demnf)
   egen demncz=std(demnc)


TITLE "SAS Combined Model Predicting Cognition: Z-Score version";
PROC GLM DATA=work.Example5b NAMELEN=100;
MODEL cognitionz = agez gripz sexMWz demNFz demNCz
                   / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
ESTIMATE "Mean Diff: Future vs. Current"   demNFz -1 demNCz  1; * B5-B4;
RUN; QUIT; TITLE;

display "STATA Combined Model Predicting Cognition -- Z-Score Version"
regress cognitionz c.agez c.gripz c.sexmwz c.demnfz c.demncz, level(95)
lincom c.demnfz*-1 + c.demncz*1 // Mean Diff: Future vs. Current = B5-B4
```

```
SAS Combined Model Predicting Cognition: Z-Score version

R-Square     Coeff Var      Root MSE    cognitionz Mean
0.277329    5.50869E17      0.853998         1.5503E-16

                                    Standard
Parameter          Estimate            Error    t Value    Pr > |t|      95% Confidence Limits
Intercept     -.0000000000       0.03641460      -0.00      1.0000    -.0715304533 0.0715304533
agez          -.1266426155       0.03711163      -3.41      0.0007    -.1995422637 -.0537429672
gripz         0.1640160832       0.04065637       4.03      <.0001    0.0841533715 0.2438787950
sexMWz        -.1640048976       0.03997385      -4.10      <.0001    -.2425268963 -.0854828988
demNFz        -.2077549718       0.03700100      -5.61      <.0001    -.2804372967 -.1350726469
demNCz        -.3986410493       0.03683489     -10.82      <.0001    -.4709970847 -.3262850138
Fut vs. Cur    -0.19088608       0.04836608      -3.95      <.0001     -0.28589322  -0.09587893
```

For models without interaction terms, you can get the same results by z-scoring all variables and using SAS GLM or STATA REG (which also provides std slopes for linear combinations).

Comparison of Bivariate (Zero-Order) vs. Unique Effects: Note that the effect sizes in the "unique effects" solution differ slightly across methods: given in an un-squared metric by standardized slopes, d and r computed from t, and semipartial r; as given in a squared metric by r^2 computed from t, and semipartial r^2. Semi-partial R^2 values translate directly into unique contributions to the model R^2, but they are not applicable to pairwise group contrasts given three or more groups. Standardized slopes are more commonly reported, but they do not translate unambiguously to interaction terms.

Bottom line: there is no single "correct" version of effect size that is accepted universally, so just be clear about what effect size you are reporting and how it was provided to you. Providing syntax as supplemental material would be really helpful!

Table 2: Bivariate and Significant Regression Coefficients (all values p < .0001)

	Bivariate Effects			
Slope	**Est**	**SE**	**d**	**r**
Age (0=85 yrs)	-0.546	0.135	-0.346	-0.171
Grip (0=9 lbs)	0.891	0.153	0.498	0.242
Sex (0=Men)	-5.269	0.926	-0.486	-0.236
None vs Future Dementia	-5.675	1.080	-0.449	-0.219
None vs Current Dementia	-16.388	1.621	-0.865	-0.397
Future vs Current Dementia	-10.713	1.815	-0.505	-0.245

	Unique Effects							
Slope	**Est**	**SE**	**Std**	**d**	**r**	**sr**	**r^2**	**sr^2**
Age (0=85 yrs)	-0.406	0.119	-0.127	-0.293	-0.145	-0.124	0.021	0.016
Grip (0=9 lbs)	0.604	0.150	0.164	0.346	0.170	0.147	0.029	0.022
Sex (0=Men)	-3.657	0.891	-0.164	-0.352	-0.173	-0.150	0.030	0.022
None vs Future Dementia	-5.722	1.019	-0.208	-0.481	-0.234		0.055	
None vs Current Dementia	-16.480	1.523	-0.399	-0.928	-0.421		0.177	
Future vs Current Dementia	-10.758	1.708	-0.191	-0.540	-0.261		0.068	

Note: In bivariate effects, r = Pearson correlation (which was converted to d). In unique effects, $Std = Est\left(\frac{SD_x}{SD_y}\right)$, $d = \frac{2t}{\sqrt{DF_{den}}}$; $r = \frac{t}{\sqrt{t^2+DF_{den}}}$; sr = semi-partial correlation from slope-specific sums of squares over corrected total sums of squares.

Example Results Section Using SAS Output [notes about what also to include]:

We estimated a series of general linear models to examine the extent to which cognition could be predicted from age, grip strength, sex (0 = men, 1= women). Table 1 provides descriptive statistics and Pearson bivariate correlations among these variables, in which cognition was predicted to be significantly greater in persons who were younger, weaker, and in women. To provide a meaningful intercept in the models that follow, we centered age such that 0 = 85 years and grip strength such that 0 = 9 pounds per square inch.

Table 1: Descriptive Statistics and Bivariate Correlations (bold values indicate *p* < .0001)

	Cognition	Age	Grip	Sex (W=1)
Cognition	1.000			
Age	**−0.170**	1.000		
Grip Strength	**0.242**	**−0.184**	1.000	
Sex (W=1)	**−0.236**	0.046	**−0.403**	1.000
Mean	24.822	84.927	9.113	0.587
SD	10.989	3.430	2.983	0.493
Min	0.000	80.016	0.000	0.000
Max	44.000	96.967	19.000	1.000

In separate linear regressions, we examined the potential for a quadratic effect of age and for a quadratic effect of grip strength. Neither quadratic slope was significant [could give *p*-values or effect sizes for completeness], indicating that linear slopes for age and grip strength were likely to be sufficient. The linear age slope was significantly negative, indicating lower predicted cognition for older persons. The linear grip strength slope was significantly positive, indicate higher predicted cognition for stronger persons. In an analysis of variance, we examined the omnibus effect of dementia group (none = 42.00%, future = 12.55%, or current = 4.18%) on cognition. We found significant mean differences in cognition across the three groups, $F(2,547) = 58.52$, MSE = 99.84, $p < .0001$, $R^2 = .176$ (see Table 2 for pairwise group effect sizes). Relative to the reference group, which was the none group (M = 27.198, SE = 0.500), cognition was significantly lower by 5.675 (SE = 1.0799) in the future group (M = 21.522, SE = 0.957) and significantly lower by 16.388 (SE = 1.621) in the current group (M = 10.810, SE = 1.542). Cognition in the current group was also significantly lower than the future group by 10.713 (SE = 1.815).

Finally, we combined the effects of all four predictors in the same model (i.e., an analysis of covariance or multiple linear regression). Semipartial eta-squared (η^2) effect sizes were obtained to describe the amount of variance captured by distinct sets of predictor slopes. We found that the unique contribution of each predictor remained significant in the same direction as their bivariate effects. The model accounted for a significant reduction in the cognition variance, $F(5,544) = 41.75$, MSE = 88.07, $p < .0001$, $R^2 = .277$, and the omnibus effect of dementia group remained significant, $F(2,544) = 67.06$, MSE = 88.07, $p < .0001$, semipartial $\eta^2 = .178$. The demographic variables of age and sex had a smaller but significant unique joint contribution, $F(2,544) = 13.79$, MSE = 88.07, $p < .0001$, semipartial $\eta^2 = .034$, indicating that controlling for dementia status and grip strength did not remove their effects.

Table 2 provides a comparison of the bivariate and unique effects [in practice you should pick only one version of the effect sizes and explain how it was provided to you]. [Emphasize why it matters based on your research questions that the predictors had significant unique effects.]